

(*The un-normalized Gaussian is given by:*)


```
g[x_] := Exp[-x^2]
```

g[x]

e^{-x^2}

(*It looks like this:*)

```
Plot[g[x], {x, -3, 3}]
```



(*The area under the curve is:*)

```
area = Integrate[g[x], {x, -Infinity, Infinity}]
```

$\sqrt{\pi}$

(*Note this is equal to Euler Gamma[1/2]:*)

```
Gamma[1 / 2]
```

$\sqrt{\pi}$

(*We can define a normalized Gaussian with area one as:*)


```
G[x_] := g[x] / Sqrt[Pi]
```

G[x]

$\frac{e^{-x^2}}{\sqrt{\pi}}$

(*It looks like this:*)

```
Plot[G[x], {x, -3, 3}]
```



(*The area under the curve is:*)

```
Integrate[G[x], {x, -Infinity, Infinity}]
```

```
1
```

(*We can estimate the width as the distance between the inflection points, which we find by setting the second derivative to zero:*)

```
Solve[D[D[G[x], x], x] == 0, x]
```

```
{{x -> -1/Sqrt[2]}, {x -> 1/Sqrt[2]}}
```

(*The standard normal distribution is defined to have both area and width of one:*)

```
phi[x_] := Exp[-x^2/2] / Sqrt[2*Pi]
```

```
phi[x]
```

$$\frac{e^{-\frac{x^2}{2}}}{\sqrt{2\pi}}$$

```
Integrate[phi[x], {x, -Infinity, Infinity}]
```

```
1
```

```
Solve[D[D[phi[x], x], x] == 0, x]
```


```
{{x -> -1}, {x -> 1}}
```

(*The erf function is:*)

```
erf[x_] := Integrate[2*G[t], {t, 0, x}]
```


(*This is a built in function in Mathematica called Erf[x]. Note that Erf[x] is antisymmetric or odd with special values Erf[-∞]=-1, Erf[0]=0, Erf[∞]=1. The plot is:*)

```
Plot[Erf[x], {x, -3, 3}]
```



(*This is just the area under the curve of G from -x to x:*)


```
Plot[G[x], {x, -1, 1}, PlotRange -> {0, 1}]
```



(*The complementary error function is Erfc with $\text{Erfc}[-\infty]=2$, $\text{Erfc}[0]=1$, $\text{Erfc}[\infty]=0$ *)




`Plot[Erfc[x], {x, -3, 3}]`



(*The imaginary error function is Erfi with $\text{Erfi}[-\infty] = -\infty$, $\text{Erfi}[0] = 0$, $\text{Erfi}[\infty] = \infty$ and the plot looks like:*)



`Plot[Erfi[x], {x, -3, 3}]`



(*As in the textbook we can expand $\text{Erf}[x]$ in a Taylor Series about $x=0$, keeping the first 10 terms:*)

`Series[Erf[x], {x, 0, 10}]`

$$\frac{2x}{\sqrt{\pi}} - \frac{2x^3}{3\sqrt{\pi}} + \frac{x^5}{5\sqrt{\pi}} - \frac{x^7}{21\sqrt{\pi}} + \frac{x^9}{108\sqrt{\pi}} + O[x]^{11}$$

(*This is the approximation for $x \ll 1$.*)

(*The asymptotic series in the book can be considered a Taylor series about ∞ as:*)


`Series[Erf[x], {x, Infinity, 10}]`

$$1 + e^{-x^2} \left(-\frac{1}{\sqrt{\pi}x} + \frac{1}{2\sqrt{\pi}x^3} - \frac{3}{4\sqrt{\pi}x^5} + \frac{15}{8\sqrt{\pi}x^7} - \frac{105}{16\sqrt{\pi}x^9} + O\left[\frac{1}{x}\right]^{11} \right)$$

(*This is Boas Eq.10.4 with some rearrangement of terms. Use this series for $x \gg 1$.*)

(*Compare to Taylor series of $\text{ArcTanh}[x]$ about zero:*)

```
Plot[ArcTan[x], {x, -5, 5}]
```



```
Series[ArcTan[x], {x, 0, 10}]
```

$$x - \frac{x^3}{3} + \frac{x^5}{5} - \frac{x^7}{7} + \frac{x^9}{9} + O[x]^{11}$$

(*There is an asymptotic series for this also about ∞ that is:*)

```
Series[ArcTan[x], {x, Infinity, 10}]
```

$$\frac{\pi}{2} - \frac{1}{x} + \frac{1}{3x^3} - \frac{1}{5x^5} + \frac{1}{7x^7} - \frac{1}{9x^9} + O\left[\frac{1}{x}\right]^{11}$$

(*Finally we can apply this to the Euler Gamma to get:*)

```
FullSimplify[Series[Gamma[p + 1], {p, Infinity, 3}]]
```

$$e^{(-1 - \text{Log}\left[\frac{1}{p}\right]) p + O\left[\frac{1}{p}\right]^4} \left(\sqrt{2\pi} \sqrt{p} + \frac{\sqrt{\frac{\pi}{2}}}{6\sqrt{p}} + \frac{\sqrt{\frac{\pi}{2}}}{144 p^{3/2}} - \frac{139 \sqrt{\frac{\pi}{2}}}{25920 p^{5/2}} + O\left[\frac{1}{p}\right]^{7/2} \right)$$

(*With some simplification this also reduces to Boas Eq.11.5, which are the first few terms of the asymptotic series expansion for Sterling's Formula. We can pick out just the first term and simplify it:*)

```
FullSimplify[Exp[(-1 - Log[1/p]) p] * Sqrt[2 * Pi * p], p > 0]
```

$$e^{-p} p^{\frac{1}{2} + p} \sqrt{2\pi}$$

(*With a bit of algebra this is equal to p^p

$$e^{-p} \sqrt{2\pi p} \text{ that is just Eq.11.1 or Sterling's Formula.}*)$$